/****************************************************************************
**
** Copyright (C) 2011 Nokia Corporation and/or its subsidiary(-ies).
** All rights reserved.
** Contact: Nokia Corporation (qt-info@nokia.com)
**
** This file is part of the Qt Components project.
**
** $QT_BEGIN_LICENSE:BSD$
** You may use this file under the terms of the BSD license as follows:
**
** "Redistribution and use in source and binary forms, with or without
** modification, are permitted provided that the following conditions are
** met:
**   * Redistributions of source code must retain the above copyright
**     notice, this list of conditions and the following disclaimer.
**   * Redistributions in binary form must reproduce the above copyright
**     notice, this list of conditions and the following disclaimer in
**     the documentation and/or other materials provided with the
**     distribution.
**   * Neither the name of Nokia Corporation and its Subsidiary(-ies) nor
**     the names of its contributors may be used to endorse or promote
**     products derived from this software without specific prior written
**     permission.
**
** THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
** "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
** LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
** A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
** OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
** SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
** LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
** DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
** THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
** (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE
** OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE."
** $QT_END_LICENSE$
**
****************************************************************************/
//![0]
import QtQuick 1.1
import Qt.labs.components.native 1.0
import "UIConstants.js" as UI

FlickrPage {
    id: thumbnailPage

    property XmlListModel model
    property bool inPortrait

    signal photoClicked(string url, int photoWidth, int photoHeight,
                        string author, string date, string description,
                        string tags, string title)

    Loader {
        sourceComponent: inPortrait ? listComponent : gridComponent
        anchors { fill: parent; margins: UI.LISTVIEW_MARGIN }
    }

    Component {
        id: gridComponent

        GridView {
            property int thumbnailsInRow: 4

            function cellWidth() {
                return Math.floor(width / thumbnailsInRow);
            }

            cacheBuffer: 2 * height
            cellHeight: cellWidth
            cellWidth: cellWidth()
            delegate: GridDelegate {
                onPhotoClicked: {
                    thumbnailPage.photoClicked(url, photoWidth, photoHeight, author,
                                               date, description, tags, title);
                }
            }
            model: thumbnailPage.model

            onWidthChanged: {
                thumbnailsInRow = width / (UI.THUMBNAIL_WRAPPER_SIDE + UI.THUMBNAIL_SPACING);
            }
        }
    }

    Component {
        id: listComponent

        ListView {
            cacheBuffer: 2 * height
            delegate: ListDelegate {
                onPhotoClicked: {
                    thumbnailPage.photoClicked(url, photoWidth, photoHeight, author,
                                               date, description, tags, title);
                }
            }
            model: thumbnailPage.model
        }
    }
}
//![0]
